# Differentialgleichungen

## Weitere Beispiele zur Separation der Variablen

In[45]:=
```
DirectionField[DE_, y_[x_], {x_, a_, b_}, {y_, c_, d_}, options___] := Module[{g},
  g = DE[[2]] /. y[x] → y;
  VectorPlot[{1, g}, {x, a, b}, {y, c, d}, options]
 ]
```

- **Beispiel 1.21**

```
DE = y'[x] == x y[x]^2
```

$y'(x) = x\, y(x)^2$

```
plot1 = DirectionField[DE, y[x], {x, -5, 5}, {y, -1, 1}, Frame → True]
```




- **Einige Lösungen**

```
DSolve[{DE, y[0] == 0}, y[x], x]
```

DSolve::bvnul : For some branches of the general solution, the given boundary conditions lead to an empty solution. ≫

{}

```
DSolve[{DE, y[x0] == y0}, y[x], x]
```

$$\left\{\left\{y(x) \to -\frac{2\,\text{y0}}{x^2\,\text{y0} - \text{x0}^2\,\text{y0} - 2}\right\}\right\}$$

```
lösung = y[x] /. DSolve[{DE, y[0] == k}, y[x], x][[1]]
```

$$-\frac{2\,k}{k\,x^2 - 2}$$

```
liste = Table[lösung, {k, -3, 3, 1/3}]
```

$$\left\{\frac{6}{-3x^2-2}, \frac{16}{3\left(-\frac{8x^2}{3}-2\right)}, \frac{14}{3\left(-\frac{7x^2}{3}-2\right)}, \frac{4}{-2x^2-2}, \frac{10}{3\left(-\frac{5x^2}{3}-2\right)}, \frac{8}{3\left(-\frac{4x^2}{3}-2\right)}, \frac{2}{-x^2-2}, \frac{4}{3\left(-\frac{2x^2}{3}-2\right)}, \frac{2}{3\left(-\frac{x^2}{3}-2\right)}, 0, -\frac{2}{3\left(\frac{x^2}{3}-2\right)}, -\frac{4}{3\left(\frac{2x^2}{3}-2\right)}, -\frac{2}{x^2-2}, -\frac{8}{3\left(\frac{4x^2}{3}-2\right)}, -\frac{10}{3\left(\frac{5x^2}{3}-2\right)}, -\frac{4}{2x^2-2}, -\frac{14}{3\left(\frac{7x^2}{3}-2\right)}, -\frac{16}{3\left(\frac{8x^2}{3}-2\right)}, -\frac{6}{3x^2-2}\right\}$$

```
plot2 = Plot[liste // Evaluate, {x, -5, 5}, PlotStyle → Thickness[0.003]]
```




```
Show[plot1, plot2, PlotRange → {-1, 1}]
```




- **Beispiel 1.22**

```
DE = y'[x] == y[x]^2 - a
```

$y'(x) = y(x)^2 - a$

```
DSolve[DE, y[x], x]
```

$\left\{\left\{y(x) \to -\sqrt{a}\, \tanh\left(\sqrt{a}\, x - \sqrt{a}\, c_1\right)\right\}\right\}$

```
DSolve[DE /. {a → -1}, y[x], x]
```

$\{\{y(x) \to \tan(c_1 + x)\}\}$

```
DSolve[DE /. {a → 1}, y[x], x]
```

$$\left\{\left\{y(x) \to \frac{1 - e^{2 c_1 + 2 x}}{e^{2 c_1 + 2 x} + 1}\right\}\right\}$$

```
DSolve[DE /. {a → 0}, y[x], x]
```

$$\left\{\left\{y(x) \to \frac{1}{-c_1 - x}\right\}\right\}$$

## ■ Beispiel 1.23

$$\text{DE} = y'[x] == x \frac{2\, y[x]^2 - 2\, y[x] - 4}{2\, y[x] - 1}$$

$$y'(x) = \frac{x\,(2\, y(x)^2 - 2\, y(x) - 4)}{2\, y(x) - 1}$$

```
plot1 = DirectionField[DE, y[x ], {x, -5, 5}, {y, -5, 5}, Frame → True]
```




## ■ Einige Lösungen

```
lösung = y[x] /. DSolve[{DE, y[0] == k}, y[x], x]
```

Solve::ifun : Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

Solve::ifun : Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

$$\left\{\frac{1}{2}\left(1 - \sqrt{9 - \left(-4 k^2 + 4 k + 8\right) e^{x^2}}\right), \frac{1}{2}\left(\sqrt{9 - \left(-4 k^2 + 4 k + 8\right) e^{x^2}} + 1\right)\right\}$$

```
y[x] /. DSolve[{DE}, y[x], x]
```

$$\left\{\frac{1}{2}\left(1 - \sqrt{9 - 4 e^{c_1 + x^2}}\right), \frac{1}{2}\left(\sqrt{9 - 4 e^{c_1 + x^2}} + 1\right)\right\}$$

```
liste1 = Table[lösung[[1]], {k, -3, 3, 1/3}];
```

```
plot2 = Plot[liste1 // Evaluate, {x, -5, 5}, PlotStyle → Thickness[0.003]]
```




```
liste2 = Table[lösung[[2]], {k, -3, 3, 1/3}];
```

```
plot3 = Plot[liste2 // Evaluate, {x, -5, 5}, PlotStyle → Thickness[0.003]]
```




```
Show[plot1, plot2, plot3, PlotRange → {-5, 5}]
```



### ■ Beispiel 1.24

In[46]:=
```
DE = y'[x] == Sin[(x + y[x])/2] - Sin[(x - y[x])/2]
```

Out[46]= $y'(x) = \sin\left(\frac{1}{2}(y(x)+x)\right) - \sin\left(\frac{1}{2}(x - y(x))\right)$

In[47]:=
```
plot1 = DirectionField[DE, y[x], {x, -5, 5}, {y, -5, 5}, Frame → True]
```

Out[47]=



■ **Ältere Versionen von *Mathematica* finden keine Lösungen, aber *Mathematica* 5.2:**

In[48]:=
```
lösung = y[x] /. DSolve[{DE, y[0] == k}, y[x], x]
```

Solve::ifun : Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

Solve::ifun : Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

Out[48]= $\left\{4\cot^{-1}\left(\frac{e^{-2\sin\left(\frac{x}{2}\right)}}{\sqrt{\tan^2\left(\frac{k}{4}\right)}}\right)\right\}$

■ ***Mathematica* kann auch den Integranden vereinfachen:**

In[49]:=
```
DE = Map[TrigExpand, DE]
```

Out[49]= $y'(x) = 2\cos\left(\frac{x}{2}\right)\sin\left(\frac{y(x)}{2}\right)$

In[50]:=
```
DSolve[DE, y[x], x]
```

Solve::ifun : Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

Out[50]= $\left\{\left\{y(x) \to 4\cot^{-1}\left(e^{-\frac{c_1}{2}-2\sin\left(\frac{x}{2}\right)}\right)\right\}\right\}$

- **Wir lösen die Differentialgleichung schrittweise:**

In[51]:=
```
gleichung = ∫ 1/Sin[y/2] ⅆy == ∫ Cos[x/2] ⅆx
```

Out[51]= $2\log\left(\sin\left(\frac{y}{4}\right)\right) - 2\log\left(\cos\left(\frac{y}{4}\right)\right) = 2\sin\left(\frac{x}{2}\right)$

In[52]:=
```
Solve[gleichung, y]
```

Solve::ifun : Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

Out[52]= $\left\{\left\{y \to 4\cot^{-1}\left(e^{-\sin\left(\frac{x}{2}\right)}\right)\right\}\right\}$